
Set Lower Limits and Read (in CL) 
 
Though it is very common to use SETLL and READ in an RPG program to retrieve a record from a 
database file, there is a similar mechanism to get a record within a CL program.  The CL equivalent of 
SETLL and READ is OVERDBF and RCVF. 
 
             PGM        PARM(&LIN &DPT) 
 
             DCL        VAR(&LIN) TYPE(*CHAR) LEN(1) 
             DCL        VAR(&DPT) TYPE(*CHAR) LEN(2) 
             DCL        VAR(&LINNBR) TYPE(*DEC) LEN(1 0) 
             DCL        VAR(&DPTNBR) TYPE(*DEC) LEN(2 0) 
             DCL        VAR(&DPTVAR) TYPE(*CHAR) LEN(2) 
             DCL        VAR(&FILEKEY) TYPE(*CHAR) len(3) 
             DCLF       FILE(LFSDPTC1) 
 
             CHGVAR     VAR(&FILEKEY) VALUE(&LIN || &DPT) 
             CHGVAR     VAR(&LINNBR) VALUE(&LIN) 
             CHGVAR     VAR(&DPTNBR) VALUE(&DPT) 
 
             OVRDBF     FILE(LFSDPTC1) POSITION(*KEYAE 2 *N &FILEKEY) 
             RCVF 
             MONMSG     MSGID(CPF0864) EXEC(GOTO CMDLBL(EXITPGM)) 
 
             CHGVAR     VAR(&DPTVAR) VALUE(&SDDPT) 
             IF         COND(&SDDPT *NE &DPTNBR) THEN(SNDUSRMSG + 
                          MSG('Department code ' || &DPT |> 'is not + 
                          on file.') MSGTYPE(*INFO)) 
             ELSE       CMD(SNDUSRMSG MSG('Record received for + 
                          department' |> &DPTVAR |> &SDDPTN) + 
                          MSGTYPE(*INFO)) 
 
 EXITPGM: 
             DLTOVR     FILE(LFSDPTC1) 
             MONMSG     MSGID(CPF0000) 
             RETURN 
              
Fig. 1 
 
The sample program in Fig. 1 illustrates the technique.  In this example the CL is attempting to retrieve 
the shipping variables at the department level. The key to the view (LFSDPTC1) is line number and 
department code, a single-digit and a two-digit numeric field, respectively. 
 
The input to the test program is a single-byte alpha field, and a two-byte alpha field, representing the 
line number and department code. After the program variables are declared, the file is declared (DCLF).  
 
In order to retrieve the correct record for the program, a variable to contain the file keys will have to be 
created. In this case a three-character variable is assigned the value of the line number concatenated 
with the department code. Note: even though line number and department code are defined as numeric 
entries, the key being created is a character value.  
 



Having constructed a file key, the OVRDBF statement is issued. The override functions similar to a SETLL 
statement when the position parameters are used. In this case POSITION is assigned to reflect the 
stating position in the file. The value *KEYAE specifies that the record identified by the search values is 
to be the first record read.  If there is no record on file that matches the file key values, the record with 
the next highest value is selected. 
 
Since *KEYAE  has been specified the number of key fields should be entered as well. In this case 2 key 
fields will be used to position the file. The number of key fields entered does not have to correspond to 
the actual number of key fields on the file, there may be three, or four, but the positioning will only use 
the first two fields. For example, if you specify POSITION(*KEY  1  RECFMTB  A), the first record in the file 
format RECFMTB that has a first key field value of A is read.  If you specify a key value of zero, the search 
is based on all key fields.  If zero is used, the key value must contain the maximum key size.  If it does 
not, no match occurs.  
 
Multiple format file can be processed by specifying the name of the record format to select (read from), 
however, in this case *N signals that no record format name is specified, all record formats (in this case 
only one) are searched for the first record that matches the other search values regardless of record 
format.                                  
 
The key value that was created by concatenating the line number and the department code indicated 
the first record to be read.  The key value is may be specified as a character string enclosed in 
apostrophes for character or positive zoned decimal formats, or as a character variable (as in Fig.1) or it 
may be specified in hexadecimal form. (For example: X’123F’ for a packed decimal representation of 
123.) 
 
In this program (Fig.1) the instruction to read a record is RCVF (receive file). Notice there is a monitor 
message on the command. The message CPF0864 is the end of file notice that will occur when no more 
records can be retrieved.  
 
If a record has been received and returned to the program, the value of the department code returned 
is tested against the value of the department code requested. Just as in an RPG program when a SETLL 
operation is followed by a READ, the information returned must be examined to determine whether or 
not it was the data requested. 
 
Keep in mind, the position operation using *KEYAE will position the file to read the record requested, if 
the key value is present on the file. If the key value is not on file, the RCVF operation will return the 
record with the next highest key value.  
 

1. If the parameters entered were ‘1’ and ‘02’, and the department code 2 for line 1 is on file, the 
first record retrieved will be the record for department code 2. 

2. However, if the department codes defined for line 1 on the shipping variables department level 
file are 1,2,3,4,5,98,99 and the parameters entered were ‘1’ and ‘35’, the first record returned 
to the program will be for department code 98, since that is the next highest key value. 

